
1 Reason for Contribution

In the ARC meeting in Vancouver, it was agreed to merge the approaches put forward in input documents OMA-ARC-CBCS-2007-0035 and OMA-ARC-CBCS-2007-0036 for the specification of the CBCS-1 protocol.  This document presents a consolidated approach by the two OMA members involved.
2 Summary of Contribution

Section 3 proposes text to be included into the CBCS Technical Specifications.
In addition we propose to add normative references in the CBCS Technical Specification for IETF RFCs 2234, 2616, 3507, 3986, for the 2003 ICAP Extensions (IETF) and for ISO 3166-2.
3 Detailed Proposal

5.1 CBCS-1 Abstract Protocol
The CBCS-1 interface allows a Content Categorization Requestor to request a Content Category (or Content Categories) for given Content (or a reference to Content) from a Content Categorization Component.  The protocol on this interface is essentially a request-response protocol that carries Content or a Content reference in the request, and a list of zero or more Content Categories in the response.

The approach to the specification of CBCS-1 is the following: this section provides an abstract, implementation neutral specification of the CBCS-1 messages. That is, it specifies only the CBCS-1 request and response message parameters but not the request-response protocol used to transfer these, nor the message encoding scheme that is used.

Section 5.2 then specifies how the Internet Content Adaptation Protocol (ICAP) [RFC3507] can transfer the CBCS-1 request and response parameters.  Though this Technical Specification does not specify encodings of the CBCS-1 messages in terms of other request-response protocols than ICAP, future versions of the CBCS Enabler MAY include  additional CBCS-1 representations.
This section uses Augmented Backus Naur Form (ABNF) as defined in [RFC2234].

The CBCS-1 protocol is an application level protocol with two types of messages: request and response.
CBCS1-Message = CBCS1-Request / CBCS-1-Response
The following paragraphs specify the CBCS1 request and response structures in detail.

5.1.1 CBCS-1 Request
A CBCS-1 request carries either Content or a Content reference, preceded by a Content descriptor and followed by an optional filter that specifies the categorization schemes to be used.  The ABNF definition of the CBCS-1 request parameters is the following:

CBCS1-Request = content-descriptor ( content / content-reference ) [ filter ]

content-descriptor = * CHAR

content = 1* OCTET               ; the content itself consists of any sequence of octets

content-reference = content-locator / content-identifier / content-digest

content-locator = content-locator-type content-locator-value

content-locator-type = token             ; eg. URI, SMS shortcode

content-locator-value = * CHAR

content-identifier = content-identifier-type content-identifier-value

content-identifier-type = token          ; eg. ISBN, title

content-identifier-value = 1* CHAR

content-digest = content-digest-type content-digest-value

content-digest-type = token              ; eg. MD5, RIPEMD-160, thumbnail

content-digest-value = 1* OCTET

filter = 1* categorization-scheme-identifier

categorization-scheme-identifier = token ; eg. ESRB , ICRA , MPAA , MRA , PEGI , RIAA

categorization-scheme-identifier =/ 1* CHAR ; free text for extension
The Content descriptor helps the Content Categorization Component determine how to decode and process the Content.  Although it is a free text field allowing for any Content meta-data, it is RECOMMENDED to use the Content-Type element from [RFC2616] which describes the type of Content (for example “text/plain“ or “image/jpeg“). The Content-Encoding,  Content-Language and Content-Length elements defined in [RFC2616] are RECOMMENDED for use in the Content descriptor parameter as well.
The CBCS-1 request carries the Content to be categorized as a series of octets.  This allows for text encoding of binary Content, though encoding techniques are not defined in this section.  They are handled by the transfer protocol that carries the CBCS-1 parameters.
The CBCS-1 request message can also carry a Content reference rather than Content itself.  A Content reference can take three forms:

· A Content locator provides information about where or how to access the Content to be categorized.  A Content locator MAY be an URI or an SMS short code (a combination of a three or four digit short number and a keyword typically used to access WAP Push Content), or any other string value.  If the CBCS-1 request carries a Content locator, then the Content Categorization Component SHALL determine the Content Categories of the Content at the specified location.
· A Content identifier is a string that provides an unambiguous textual description of the Content. A Content identifier MAY be an ISBN number for a book or an ISAN number for audiovisual Content.  It MAY also be a movie title such as "Casablanca, 1949" or any other string value, provided it can unambiguously reference the Content.  If the CBCS-1 request carries a Content identifier, then the Content Categorization Component SHALL use this Content identifier to determine the Content Categories to be returned to the requestor.  The Content Categorization Component MAY use this Content identifier to retrieve cached Content Categories.

· A Content digest is a binary value that can be used as an unambiguous reference to Content.  A Content digest MAY be a reduced form of the Content which conserves the main properties of the Content for categorization, e.g. a thumbnail or movie with strongly reduced frame size and rate.  A Content digest MAY also be a digital digest computed on the Content, for example a MD5 or RIPEMD-160 hash.  If the CBCS-1 request carries a Content digest, then the Content Categorization Component SHALL use this Content digest to determine the Content Categories to be returned to the requestor.  If the Content digest is a digital digest (hash) then the Content Categorization Component MAY use this digital digest to retrieve cached Content Categories.

The content-reference element has specific semantics.  If the content-locator-type is URI then the content-locator-value MUST have the syntax defined in [RFC3986].  If the content-locator-type is SMS short code, then the content-locator-value MUST consist of one or more digits, which MAY be followed by a plain text keyword or message of no more than 161 characters.
If the content-identifier-type is ISBN, then the content-identifier-value MUST consist of 13 digits.  If the content-identifier-type is ISAN, then the content-identifier-value MUST consist of 24 hexadecimal digits.
The optional filter parameter in the CBCS-1 request allows the requestor to specify that it is only interested to receive Content Categories from specific categorization schemes.  It does so by simply listing the names of the categorization schemes it wants the Content Categorization Component to use.   These MAY be predefined categorization scheme identifiers, or they MAY be any string value so as to allow for the CBCS Enabler to use any categorization schemes.
The categorization-scheme-identifier element MUST support tokens for the following categorization schemes: ESRB (Entertainment Software Rating Board), ICRA (Internet Content Rating Association), MPAA (Motion Picture Association of America) , MRA (Minimum Recommended Age), PEGI (Pan European Game Information) and Recording Industry Association of America (RIAA).
5.1.2 CBCS-1 Response
The CBCS-1 response message provides a status code, a (possibly empty) vector of Content Categories, an optional signature and an optional index.  The ABNF definition of the CBCS-1 response parameters is the following:
CBCS1-Response = status-code content-category-vector [ signature ] [ index ]

status-code = *CHAR
content-category-vector = * ( content-category [ * region-code ] )

content-category = [ categorization-scheme-identifier ] content-category-value

content-category-value = * CHAR

region-code = 2 CHAR       ; two characters, conforming to [ISO3166-2]

signature = signature-type signature-value

signature-type = token       ; eg. DSA, ECDSA

signature-value = * OCTET


index = content-identifier / content-digest ; content-identifier and content-digest as above

The status-code element provides the requestor with information about the status of the response.  The status-code element SHALL contain one of the CBCS-1 status codes listed in table 1 below.

Table 1. CBCS-1 specific status codes

	Normal response

	200
	OK

	Client errors

	400
	Bad request

	440
	Badly formed filter

	441
	Corrupt or incomplete content

	442
	Unable to resolve content reference

	443
	No content at specified location

	Server errors

	500
	Server error

	550
	Server does not support requested categorization scheme

	551
	Server does not support content type

	552
	Server does not support content encoding


The content-category-vector element contains the Content Categories (zero or more) returned to the requestor.   The CBCS Enabler supports a number of widely used categorization schemes, which are specified in Appendix E (normative).  The CBCS-1 response MAY carry also any other type of Content Category (not specified in Appendix E) in plain text format.

The optional signature argument in the CBCS-1 response carries a digital signature of the Content Categorization Component calculated on the Content and its Content Categories.  The requestor MAY use this signature to create trusted Pre-Categorized Content, while a Content Screening Component MAY use this signature to verify the authenticity and integrity of Pre-Categorized Content it receives.
The optional index element MAY be used by the Content Categorization Component if it caches the categorization result.  It allows future requests to be made with the index value alone, rather than with the complete Content.  This can considerably improve the efficiency of the CBCS-1 protocol if categorization requests occur often for a piece of Content.
5.2 ICAP representation of CBCS-1

This section defines how the CBCS-1 request and response parameters can be transferred using a widely accepted protocol for Content vectoring, the Internet Content Application Protocol (ICAP) [RFC3507].
The ICAP representation CBCS-1 messages defined in this section adheres to the following principles:

· If the Content to be categorized is a HTTP request or HTTP response, then it MUST be possible to use ICAP as described in [RFC3507].

· If the Content to be categorized is not a HTTP request or HTTP response, then the content or content‑reference element in a CBCS-1 request SHALL be transferred in the encapsulated body of the resulting ICAP message.
· CBCS-1 parameters other than content or content-reference SHALL be transferred in ICAP headers. Where necessary, new user-defined X-extension headers are introduced for these parameters.

5.2.1 Transfer of the CBCS-1 Request by ICAP

A CBCS1-request message SHALL be encoded in an ICAP REQMOD or RESPMOD method [RFC3507].
If the CBCS1-request contains a content element representing a HTTP request or HTTP response, then:

· If the content element consists of a HTTP request, then it SHALL be encapsulated in an ICAP REQMOD method as specified in [RFC3507].
· If the content element consists of a HTTP response, then it SHALL be encapsulated in an ICAP RESPMOD method as specified in [RFC3507], using chunk-encoding to encapsulate the body of the HTTP response.

· The CBCS-1 content-descriptor element SHOULD be encoded as one or more [RFC2616] entity-header elements in the encapsulated header section of the ICAP message, e.g. if it comes in the form of a Content-Type or Content-Language entity header as recommended in section 5.1.1.
· The CBCS-1 content-descriptor element MAY be transferred in the user-defined X-Content-Descriptor extension header if it cannot be expressed as [RFC2616] entity-header. The syntax of the X-Content-Descriptor extension header is:
X-Content-Descriptor = “X-Content-Descriptor:” content-descriptor 

                                     ; content-descriptor as defined in section 5.1.1.
If the CBCS1-request does not contain a content element representing a HTTP request or HTTP response, then:

· The content or content-reference element of the CBCS1-request SHALL be transferred in the encapsulated body part of an ICAP RESPMOD message, using chunk encoding.  
· If the CBCS1-request contains a content element, then it SHOULD be encoded as a single chunk.    

· If the CBCS1-request contains a content-reference element, then it SHALL be encoded in two chunks.  The first chunk SHALL contain the locator-type, identifier-type or digest-type and the second chunk SHALL contain the locator-value, identifier-value or digest-value.

· The content-descriptor element of the CBCS1-request SHOULD be transferred in the chunk encoded encapsulated body [RFC3507] of the ICAP message, e.g. if it comes in the form of a Content-Type or Content-Language entity header as recommended in section 5.1.1.
· The content-descriptor element of the CBCS1-request MAY be transferred in the user-defined X-Content-Descriptor extension header if it cannot be expressed by the RECOMMENDED Content-Type, Content-Encoding, Content-Language or Content-Length elements [RFC2616] in the chunk encoded encapsulated body of the ICAP message.
The CBCS-1 filter element SHALL be transferred in the user-defined X-Filter extension header. The syntax of the X-Filter extension header is:
X-Filter = “X-Filter:” categorization-scheme-identifier 

                   * ( separator categorization-scheme-identifier) 

                                      ; categorization-scheme-identifier is a token or string 

                                      ; as defined in section 5.1.1

separator = token          ; for example comma

Note that this codification of the CBCS-1 request in an ICAP request message fully conforms to [RFC3507].
5.2.2 Transfer of the CBCS-1 Response by ICAP
A CBCS1-response message SHALL be encoded in an ICAP response in the following way. 
The ICAP status code SHALL match the CBCS-1 status-code element (see table 1 in section 5.1.2).  
The X-Attribute ICAP extension header SHALL contain the CBCS-1 content-category-vector element if and only if it is not null.  The elements of the content-category-vector MUST be comma separated.  The X-Attribute ICAP extension header MUST be omitted if the content-category-vector element is null, to comply with [ICAP-Extensions].

The optional CBCS-1 signature element SHALL be transferred in the user-defined X-Signature extension header.  The syntax of the X-Signature extension header is:
X-Signature = “X-Signature:” signature-type separator signature-value

separator = token          ; for example comma

signature-type = token     ; as defined in section 5.1.2
signature-value = * OCTET  ; as defined in section 5.1.2
The optional CBCS-1 index element SHALL be transferred in the encapsulated body part of the ICAP response message, as it represents a modified (reduced) form of the original Content submitted in the request.  If the CBCS-1 index element is not present, then the ICAP encapsulated body SHALL be the null-body [RFC3507].

The optional X-Response-Desc ICAP extension header [ICAP-Extensions] MAY be used to provide the requestor with additional information about the response.  In case the ICAP status code is 200, the X-Response-Desc extension header MAY contain:

· The string “categorized” if the CBCS-1 content-category-vector element is non-empty and the CBCS-1 signature and index elements are not present.  
· The string “categorized with index” if the CBCS-1 content-category-vector element is non-empty, the CBCS-1 index element is present and the CBCS-1 signature element is not present.
· The string “categorized with signature” if the CBCS-1 content-category-vector element is non-empty, the CBCS-1 signature element is present and the CBCS-1 index element is not present.
· The string “categorized with index and signature” if the CBCS-1 content-category-vector element is non-empty, and both the CBCS-1 signature and index elements are present.

Note that this codification of the CBCS-1 response in an ICAP response message fully conforms to [RFC3507].
Appendix E. Categorization schemes supported by CBCS 
(NORMATIVE)

This normative appendix species the semantics of the content-category element (section 5.1.2) for supported Content categorization schemes.  The table below should be interpreted as follows: “if the categorization-scheme-identifier is as displayed in the left column, then the syntax of the content-category-value MUST have the syntax specified in the right column”.


	categorization-scheme-identifier
	content-category-value

	ESRB
	content-category-value =/ ESRB-rating [ ESRB-content-descriptor ]
ESRB-rating =    “EC”    ; Early Childhood

               / “E”     ; Everybody 

               / “E10+”  ; Everybody 10 and over

               / “T”     ; Teenagers (13 and over)

               / “M”     ; Mature (17 and over)

               / “AO”    ; Adults Only

               / “RP”    ; Rating Pending
ESRB-content-descriptor = ”Alcohol Reference”

 
/”Animated Blood”

 
/”Blood” 

 
/”Blood and Gore”

 
/”Cartoon Violence” 

 
/”Comic Mischief” 

 
/”Crude Humor” 

 
/”Drug Reference”

 
/”Fantasy” 

 
/”Intense Violence” 

 
/”Language” 

 
/”Lyrics” 

 
/”Mature Humor”

 
/”Nudity” 

 
/”Partial Nudity”

 
/”Real Gambling”

 
/”Sexual Content”

 
/”Sexual Themes”

 
/”Sexual Violence”

 
/”Simulated Gambling”

 
/”Strong Language”

 
/”Strong Lyrics”

 
/”Strong Sexual Content”

 
/”Suggestive Themes”

 
/”Tobacco Reference”

 
/”Use of Drugs”

 
/”Use of Alcohol”

 
/”Use of Tobacco”

 
/”Violence”

 
/”Violent References”

	ICRA
	content-category-value =/ 1* ICRA-label

ICRA-label =    “na 1” ; Exposed breasts 

              / “nb 1” ; Bare buttocks 

              / “nc 1” ; Visible genitals 

              / “nz 0” ; Nudity may be present 

              / “nz 1” ; No nudity
              / “sa 1” ; Passionate kissing 

              / “sb 1” ; Obscured or implied sexual acts 

              / “sc 1” ; Visible sexual touching

              / “sd 1” ; Explicit sexual language 

              / “se 1” ; Erections/explicit sexual acts 

              / “sf 1” ; Erotica 

              / “sz 0” ; Sexual material may be present 

              / “sz 1” ; No sexual material
              / “va 1” ; Assault/rape 

              / “vb 1” ; Injury to human beings 

              / “vc 1” ; Injury to animals 

              / “vd 1” ; Injury to fantasy characters
              / “ve 1” ; Blood and dismemberment, human beings 

              / “vf 1” ; Blood and dismemberment, animals 

              / “vg 1” ; Blood and dismemberment, fantasy
              / “vh 1” ; Torture or killing of human beings 

              / “vi 1” ; Torture or killing of animals 

              / “vj 1” ; Torture or killing of fantasy
              / “vz 0” ; Violence may be present 

              / “vz 1” ; No violence
              / “la 1” ; Abusive or vulgar terms 

              / “lb 1” ; Profanity or swearing 

              / “lc 1” ; Mild expletives 

              / “lz 0” ; Potentially offensive language
              / “lz 1” ; No potentially offensive language
              / “oa 1” ; Depiction of tobacco use 

              / “ob 1” ; Depiction of alcohol use 

              / “oc 1” ; Depiction of drug use 

              / “od 1” ; Depiction of the use of weapons 

              / “oe 1” ; Gambling 

              / “of 1” ; Bad example for young children 

              / “og 1” ; Fear, intimidation, horror 

              / “oh 1” ; Discrimination or hate incitement
              / “oz 0” ; Potentially harmful activities
              / “oz 1” ; No potentially harmful activities
              / “ca 1” ; User-generated content (moderated) 

              / “cb 1” ; User-generated content (unmoderated) 

              / “cz 0” ; User generated content may be present 

              / “cz 1” ; No user-generated content 

              / “xa 1” ; Artistic context 

              / “xb 1” ; Educational context 

              / “xc 1” ; Medical context 

              / “xd 1” ; Sports context 

              / “xe 1” ; News context

	MPAA
	content-category-value =/   “G”    ; General Audiences

                          / “PG”   ; Parental Guidance

                          / “PG-13”; Parents strongly cautioned

                          / “R”    ; Restricted

                          / “NC-17”; No one 17 and under admitted

	MRA
	content-category-value =/  2 DIGIT

	PEGI
	content-category-value =/ PEGI-rating [ PEGI-descriptor ]

PEGI-rating = *2 DIGIT

PEGI-descriptor = “Bad language”

                  / “Discrimination”

                  / “Drugs”

                  / “Fear”

                  / “Gambling”

                  / “Sex”

                  / “Violence”

	RIAA
	content-category-value =/  “”/ “Parental advisory”


4 Intellectual Property Rights

Members and their Affiliates (collectively, "Members") agree to use their reasonable endeavours to inform timely the Open Mobile Alliance of Essential IPR as they become aware that the Essential IPR is related to the prepared or published Specification.  This obligation does not imply an obligation on Members to conduct IPR searches.  This duty is contained in the Open Mobile Alliance application form to which each Member's attention is drawn.  Members shall submit to the General Manager of Operations of OMA the IPR Statement and the IPR Licensing Declaration.  These forms are available from OMA or online at the OMA website at www.openmobilealliance.org.

5 Recommendation

We kindly ask ARC to discuss this proposal and agree the text to be included into the CBCS Technical Specification. 

